17. Модули

Часто используемые подпрограммы хранятся в виде библиотек. Библиотеки организуются в виде модулей. Для того, чтобы использовать библиотеки (модули) достаточно задать в программе описательное предложение

uses <список имен модулей>;

Текст модуля располагается в файле, имя которого обязательно должно совпадать с именем модуля. Поэтому имя модуля не может быть длиннее 8 символов. Текст модуля транслируется в файл с расширением tpu. На этапе компоновки (сборки) задачи модуль подключается к оттранслированной программе.

Кроме собственных библиотек пользователя существуют стандартные библиотеки Tурбо Паскаля. Среди этих библиотек можно назвать следующие:

· System (ввиду исключительной важности подключается автоматически к любой программе) содержит основные подпрограммы;

· Dos обеспечивает интерфейс с программами операционной системы;

· Crt содержит средства управления монитором в текстовом режиме;

· Graph содержит средства управления монитором в графическом режиме;

· Printer обеспечивает связь с печатающим устройством;

· Overlay содержит средства для создания и управления перекрывающимися (оверлейными) программами.

V.2. Текст модуля

Существуют требования к оформлению текста модуля. Модуль определяется как совокупность описаний констант, типов, переменных, процедур и функций. Модуль состоит из секций. Синтаксис модуля определяется на рис. 17.1.

Первая секция модуля - заголовок модуля

init <имя модуля>;     .

Имя модуля должно совпадать с именем файла, в котором располагается модуль. Т.е. текст модуля должен хранится в файле  <имя модуля>.pas , оттранслированный модуль - в файле  <имя модуля>.tpu
Второй частью модуля является интерфейсная секция

interface <описания>     .

Интерфейсная секция включает в себя перечень описаний, включенных в данный модуль. Описания представляют собой описания констант, типов, переменных, заголовков подпрограмм, входящих в библиотеку.

Третьей частью модуля является секция реализации

implementation <описания подпрограмм>   .

В секцию реализации входят описания всех подпрограмм, объявленных в интерфейсной секции, а также подпрограмм, используемых при реализации библиотеки (таких подпрограмм может и не быть). В целях упрощения при описании библиотечных подпрограмм в секции реализации списки формальных параметров можно опускать.

Четвертой - необязательной секцией является секция инициализации, начинающаяся с операторной скобки begin . Секция инициализации используется в тех случаях, когда библиотечным переменным (объявленным в интерфейсной секции) необходимо при использовании библиотеки задавать одинаковые начальные значения. Таким образом, секция инициализации состоит из операторов присваивания библиотечным переменным значений констант.

Заканчивается модуль секцией завершения, которая состоит из операторной скобки end с точкой. 

Рис. 17.1. - Синтаксическая диаграмма текста модуля.

17.2. Модуль обработки комплексных чисел

В качестве примера напишем модуль, который содержит подпрограммы, обеспечивающие работу с комплексными числами.

Математическая модель

Опишем математическую модель. Пусть a, b и c - комплексные числа. Любое комплексное число может быть представлено в виде двух вещественных чисел (обозначим их Re и Im), первое из которых является действительной частью, а второе - мнимой частью комплексного числа. При реализации на Паскале логичнее представить комплексное число как запись, состоящую из двух вещественных полей Re и Im. 

Над комплексными числами можно выполнять унарные операции

получения комплексно сопряженного числа (c.Re:=a.Re; c.Im:=-a.Im),

вычисления модуля комплексного числа (|a|:=sqrt(a.Re2+a.Im2)), 

и бинарные операции вычисления

· суммы (с:=a+b; c.Re:=a.Re+b.Re; c.Im:=a.Im+b.Im);

· разности (с:=a-b; c.Re:=a.Re-b.Re; c.Im:=a.Im-b.Im);

· произведения (с:=a*b; c.Re:=a.Re*b.Re-a.Im*b.Im; c.Im:=b.Re*a.Im+a.Re*b.Im);

· - частного (с:=a/b; c.Re:=(a.Re*b.Re+a.Im*b.Im)/(b.Re2+b.Im2); c.Im:=(b.Re*a.Im-a.Re*b.Im)/(b.Re2+b.Im2) ).

Текст модуля

unit complex; {заголовок модуля}

{интерфейсная секция}

interface

type comp=record {тип комплексных данных}

            Re, {действительная часть}

            Im : real {мнимая часть}

          end;

{процедура вычисления комплексно сопряженного числа}

procedure csopr(var a,     {исходное комплексное число}

                    c :    {результат - комплексно

                            сопряженное число}

                        comp);

{функция вычисления модуля комплексного числа}

function  cmod(var a:comp  {исходное комплексное число})

               : real;

{процедура вычисления суммы двух комплексных чисел}

procedure csum(var a,b,    {исходные комплексные числа}

                    c :    {результат - c:=a+b}

                        comp);

{процедура вычисления разности двух комплексных чисел}

procedure craz(var a,b,    {исходные комплексные числа}

                    c :    {результат - c:=a-b}

                        comp);

{процедура вычисления произведения двух комплексных чисел}

procedure cumn(var a,b,    {исходные комплексные числа}

                    c :    {результат - c:=a*b}

                        comp);

{процедура вычисления частного двух комплексных чисел}

procedure cdel(var a,b,    {исходные комплексные числа}

                    c :    {результат - c:=a/b}

                        comp);

{процедура ввода комплексного числа со стандартного устройства ввода - вводимое значение задается парой вещественных констант, разделенных пробелом}

procedure creadln(var c : comp {вводимая переменная} );

{процедура вывода комплексного числа на стандартное устройство вывода - выводимое значение представляется в виде пары вещественных констант, разделенных символами  +i }

procedure cwrite(var c : comp {вводимая переменная} );

{секция реализации}

implementation

procedure csopr;

begin

  c.Re:=a.Re;

  c.Im:=-a.Im

end;{procedure csopr}

function  cmod;

begin

  cmod:=sqrt(sqr(a.Re)+sqr(a.Im))

end;{function  cmod}

procedure csum;

begin

  c.Re:=a.Re+b.Re;

  c.Im:=a.Im+b.Im

end;{procedure csum}

procedure craz;

begin

  c.Re:=a.Re-b.Re;

  c.Im:=a.Im-b.Im

end;{procedure craz}

procedure cumn;

begin

  c.Re:=a.Re*b.Re-a.Im*b.Im;

  c.Im:=b.Re*a.Im+a.Re*b.Im

end;{procedure cumn}

procedure cdel;

begin

  c.Re:=(a.Re*b.Re+a.Im*b.Im)/(sqr(b.Re)+sqr(b.Im));

  c.Im:=(b.Re*a.Im-a.Re*b.Im)/(sqr(b.Re)+sqr(b.Im))

end;{procedure cdel}

procedure creadln;

begin

  readln(c.Re,c.Im)

end{procedure cread};

procedure cwrite;

begin

  write(c.Re,'+i',c.Im)

end;{procedure cwrite}

{секция инициализации отсутствует}

{конец текста модуля}

end.

17.3. Модуль суммирования рядов

Постановка задачи и метод решения

Ряд - последовательность чисел, в которой все элементы пронумерованы, и значение элемента ряда является функцией от порядкового номера.

а1, а2, а3, ..., аi, ...      , где  аi=f(i)

Часто решается задача суммирования элементов ряда, в которой производится одно из вычислений:

- суммирование фиксированного числа элементов ряда;

- суммирование элементов бесконечного ряда.

Естественно, суммирование бесконечного ряда имеет смысл только для сходящегося ряда (при беспредельном возрастании количества элементов ряда, входящих в сумму ряда, сумма ряда стремится к определенному конечному пределу). Известно, что ряд будет сходящимся, если выполняется одно из следующих условий:

|ai+1| < |ai |  (для знакопостоянных рядов);

|ai+1 - -ai | < |ai – ai-1 | (для знакопеременных рядов).

В задаче суммирования фиксированного числа элементов ряда входной информацией должно быть количество элементов ряда, входящих в сумму.

В задачах суммирования элементов бесконечного ряда входной информацией должна быть точность, определяющая продолжение процесса суммирования Е. В этих задачах суммирование продолжается (для первого типа сходимости) пока очередной элемент по абсолютной величине превышает Е, либо (для второго типа сходимости) пока абсолютная величина разности между текущим и следующим элементами превышает Е.

Дадим определение:

Функция называется алгоритмически простой, если ее можно записать с помощью выражения на алгоритмическом языке, и алгоритмически сложной, если для реализации функции необходим некоторый алгоритм.

При решении каждой из задач суммирования рядов существует два случая - функция, определяющая элемент ряда в зависимости от порядкового номера есть

- алгоритмически простая функция;

- алгоритмически сложная функция.

Таким образом, каждый из задач суммирования имеет две разновидности.

В первом случае должна быть определена функция вычисления текущего элемента ряда в зависимости от порядкового номера, представляющая собой арифметическое выражение.

Во втором случае ряд представляется, как геометрическая прогрессия. Для этого предварительно

· вычисляется значение первого элемента. Это значение является входной информацией;

· выводится формула знаменателя геометрической прогрессии (q) подстановкой в общем виде функции определения элемента ряда в формулу:

и, если q(i) является алгоритмически простой функцией от i, то можно применить нижеуказанный алгоритм.

В каждой из задач суммирования результатом является одно вещественное число. Поэтому целесообразно оформить каждую задачу, как функцию. Цели и списки параметров для каждой функции определены. Разработаем модуль, включающий эти функции:

Текст модуля

unit sum_riad;

{интерфейсная секция}

interface

  type

  {тип функций, вычисляющих значение элемента ряда и знаменателя

   геометрической прогресси, как функций от поряд   кового номера}

  tf=function(i:integer):real;

  {функция суммирования конечного ряда с простой функцией

   определения элемента ряда}

  function z1(n{количество элементов}:integer;

            f{функция, вычисляющая значение элемента ряда в

              зависимости от порядкового номера }:tf):real;

  {функция суммирования конечного ряда со сложной функцией

   определения элемента ряда}

  function z2(n{количество элементов}:integer;

              a1{значение первого элемента ряда}:real;

              q{функция, вычисляющая значение знаменателя

                геометрической прогрессии в зависимости от

                порядкового номера }:tf):real;

  {функция суммирования бесконечного ряда первого типа

   сходимости с простой функцией определения элемента ряда}

  function z3(eps{точность}:real;

              f{функция, вычисляющая значение элемента ряда в

              зависимости от порядкового номера }:tf):real;

  {функция суммирования бесконечного ряда первого типа

   сходимости со сложной функцией определения элемента ряда}

  function z4(eps{точность}:real;

              a1{значение первого элемента ряда}:real;

              q{функция, вычисляющая значение знаменателя

                геометрической прогрессии в зависимости от

                порядкового номера }:tf):real;

  {функция суммирования бесконечного ряда второго типа

   сходимости с простой функцией определения элемента ряда}

  function z5(eps{точность}:real;

              f{функция, вычисляющая значение элемента ряда в

                зависимости от порядкового номера }:tf):real;

  {функция суммирования бесконечного ряда второго типа

   сходимости со сложной функцией определения элемента ряда}

  function z6(eps{точность}:real;

              a1{значение первого элемента ряда}:real;

              q{функция, вычисляющая значение знаменателя

                геометрической прогрессии в зависимости от

                порядкового номера }:tf):real;

{секция реализации}

implementation

  function z1;

  var i:integer;{порядковый номер элемента}

      sum:real;{сумматор - для накопления суммы}

  begin

    sum:=0;

    for i:=1 to n do  sum:=sum+f(i);

    z1:=sum

  end;{z1}

  function z2;

  var i:integer;{порядковый номер элемента}

      a:real;{текущий элемент ряда}

      sum:real;{сумматор - для накопления суммы}

  begin

    sum:=0;

    a:=a1;

    for i:=1 to n do  begin

      sum:=sum+a;

      a:=a*q(i)

    end;{for i}

    z2:=sum

  end;{z2}

  function z3;

  var i:integer;{порядковый номер элемента}

      sum:real;{сумматор - для накопления суммы}

  begin

    sum:=0;

    i:=1;

    while abs(f(i))>eps do begin

      sum:=sum+f(i);

      i:=i+1

    end;{while}

    z3:=sum

  end;{z3}

  function z4;

  var i:integer;{порядковый номер элемента}

      a:real;{текущий элемент ряда}

      sum:real;{сумматор - для накопления суммы}

  begin

    sum:=0;

    a:=a1;

    i:=1;

    while abs(a)>eps do begin

      sum:=sum+a;

      a:=a*q(i);

      i:=i+1

    end;{while}

    z4:=sum

  end;{z4}

  function z5;

  var i:integer;{порядковый номер элемента}

      ap:real;{предыдущий элемент ряда}

      at:real;{текущий элемент ряда}

      sum:real;{сумматор - для накопления суммы}

  begin

    ap:=f(1);

    sum:=ap;

    at:=f(2);

    i:=2;

    while abs(at-ap)>eps do begin

      sum:=sum+at;

      i:=i+1;

      ap:=at;

      at:=f(i)

    end;{while}

    z5:=sum

  end;{z5}

  function z6;

  var i:integer;{порядковый номер элемента}

      ap:real;{предыдущий элемент ряда}

      at:real;{текущий элемент ряда}

      sum:real;{сумматор - для накопления суммы}

  begin

    ap:=a1;

    sum:=ap;

    at:=ap*q(1);

    i:=2;

    while abs(at-ap)>eps do begin

      sum:=sum+at;

      ap:=at;

      at:=ap*q(i);

      i:=i+1

    end;{while}

    z6:=sum

  end;{z6}

end.

Пример применения модуля суммирования рядов

В качестве примера рассмотрим задачу вычисления функции sin(x). Эта функция вычисляется как сумма бесконечного ряда

Общая формула для элемента ряда имеет вид:  

Полученная функция относится к классу алгоритмически сложных, поэтому выполняем подготовительный этап:

- первый элемент ряда имеет значение x;

- функция знаменателя геометрической прогрессии

Таким образом, для суммирования бесконечно убывающего знакопеременного ряда воспользуемся подпрограммой z6 из модуля sum_riad.

В программе вычисления необходимо будет описать функцию с одним целочисленным аргументом - номером элемента ряда и результатом вещественного типа - значением знаменателя геометрической прогрессии, используемым для вычисления следующего элемента ряда.

Текст программы

program fsin;

uses sum_riad;{подключение библиотеки суммирования рядов}

var x:real;{аргумент функции}

    E:real;{точность вычисления суммы}

{$f+}

{функция вычисления знаменателя геометрической прогрессии}

function znsin(i:integer):real;

begin

  znsin:=-sqr(x)/((2*i)*(2*i+1));

end;

{$f-}

begin

  writeln('Введите точность вычисления');

  readln(E);

  writeln('Введите аргумент для функции sin');

  readln(x);

  writeln('sin(',x,')=',z6(E,x,znsin));

end.

Формальное исполнение программы

Рассмотрим процесс выполнения программы fsin.

0) До начала выполнения программы производится статическое

распределение памяти(рис. 17.2).

	
	ОП
	

	Х
	?
	Под хранение данного типа REAL

	Е
	?
	Под хранение данного типа REAL


Рис. 17.2. – Распределение ОП

Содержимое выделенных участков памяти - неопределенное;

1) выполняется обращение к процедуре вывода:

writeln('Введите точность вычисления');

На экране, начиная с текущего положения курсора, появляется сообщение:

Введите точность вычисления

, и курсор переводится в начало следующей строки экрана дисплея;

2) выполняется обращение к процедуре ввода:

readln(E);

Программа ожидает получения информации в виде одного вещественного числа с клавиатуры ПЭВМ. Человек набирает эту информацию, например, 1e-10  и нажимает клавишу ввод (Return или Enter). Значение 1e-10 поступает в память под именем E (рис.17.3);

3) выполняется обращение к процедуре вывода:

writeln('Введите аргумент для функции sin');

На экране, начиная с текущего положения курсора, появляется сообщение:

Введите аргумент для функции sin

, и курсор переводится в начало следующей строки экрана дисплея;

4) выполняется обращение к процедуре ввода:

readln(x);

Программа ожидает получения информации в виде одного вещественного числа с клавиатуры ПЭВМ. Человек набирает эту информацию, например, 0.1  и нажимает клавишу ввод (Return или Enter). Значение 0.1 поступает в память под именем x. (рис. 17.3);

	
	ОП
	

	Х
	0.1
	Под хранение данного типа REAL

	Е
	1.0Е-10
	Под хранение данного типа REAL


Рис. 17.3. – Содержимое ОП после ввода

5) выполняется обращение к процедуре вывода:

writeln('sin(',x,')=',z6(E,x,znsin));

Последовательно вычисляются фактические параметры-выражения, их символьный эквивалент в буфер вывода и выводится из буфера на экран, освобождая буфер вывода в оперативной памяти. 

Значение первого параметра - строка символов  sin. (

Значение второго параметра - вещественное число 1.0E-01 . 

Значение третьего параметра - строка символов). 

Значение четвертого параметра - вещественное число, представляющее собой результат вычисления функции z6.

Остановимся подробно на выполнении функции z6. При обращении к функции возникает динамический экземпляр функции, который с точки зрения пользователя представляет собой динамический экземпляр данных и измененный машинный код функции.

Динамический экземпляр данных (рис. 17.4) включает в себя 

· переменную, совпадающую с именем функции, под результат работы функции (z6);

· переменные, являющиеся параметрами-значениями (eps, a1);

· локальные переменные (i, ap, at, sum).

	z6
	?
	real

	eps
	?
	Real

	al
	?
	Real

	i
	?
	Integer

	ap
	?
	Real

	at
	?
	Real

	sum
	?
	Real


Рис.17.4. – Динамический экземпляр данных

После возникновения динамического экземпляра осуществляется передача фактических параметров (при этом формируются значения параметров-значений в динамическом экземпляре данных и модифицируется код подпрограммы). Вспомним заголовок подпрограммы z6:

      function z6(eps{точность}:real;

                  a1{значение первого элемента ряда}:real;

                  q{функция, вычисляющая значение знаменателя

                    геометрической прогрессии в зависимости от

                    порядкового номера }:tf):real;

и сопоставим его с обращением к функции z6:

               z6(E,x,znsin)   .

Первый и второй параметры - параметры-значения, поэтому в динамический экземпляр передаются значения фактических параметров (1.0E-10 
[image: image4.wmf]®

 eps, 0.1 

 a1). Третий параметр - параметр-переменная подпрограммного типа, поэтому происходит модификация кода функции z6. При модификации в разделе операторов функции на место формального имени q подставляется имя фактической функции znsin. Обратите внимание - заголовок функции znsin соответствует объявлению типа tf.

	z6
	?
	real

	eps
	1.0Е-10
	Real

	al
	0.1
	Real

	i
	?
	Integer

	ap
	?
	Real

	at
	?
	Real

	sum
	?
	Real


Рис.17.5. – Динамический экземпляр данных 
после передачи фактических параметров

Содержимое динамического экземпляра данных приведено на рис. 17.5 и модифицированный код:

  begin
    ap:=a1;

    sum:=ap;

    at:=ap*znsin(1);

    i:=2;

    while abs(at-ap)>eps do begin

      sum:=sum+at;

      ap:=at;

      at:=ap*znsin(i);

      i:=i+1

    end;{while}

    z6:=sum

  end;

Начинается исполнение операторов функции z6:

1) выполняется оператор присваивания

ap:=a1;

в результате ap:=0.1;

2) выполняется оператор присваивания

sum:=ap;

в результате sum:=0.1;

3) выполняется оператор присваивания

at:=ap*znsin(1);

т.к. составной частью вычисляемого выражения является обращение к функции, то подробно остановимся на вычислении этого выражения. При вычислении выражения формируется стек (будем представлять его просто участком оперативной памяти), в который помещаются значения всех операндов и знаки операций. Упрощенно для нашего случая можно сказать, что в стеке находятся число 0.1, знак умножения и результат вычисления функции znsin с фактическим аргументом 1. Рассмотрим вычисление значения функции znsin. При обращении возникает динамический экземпляр данных и передаются параметры (рис. 17.6).

	Znsin
	?
	Real

	I
	1
	Integer


Рис. 17.6. – Динамический экземпляр данных функции znsin
Программный код при этом не модифицируется, т.к. отсутствуют параметры-переменные и параметры-подпрограммы. После этого начинается выполнение программного кода функции znsin. Выполняется оператор присваивания

znsin:=-sqr(x)/((2*i)*(2*i+1));

Для вычисления выражения формируется стек, в который поступают операнды и знаки операций. При вычислении числителя дроби вычисляется функция возведения в квадрат значения переменной x. Эта переменная не принадлежит динамическому экземпляру функции znsin, поэтому переменную x необходимо искать в описании блока, в который вложена функция znsin, т.е. в программе fsin. В результате вычисления выражения -sqr(0.1)/((2*1)*(2*1+1)) получаем значение -1.667Е-03, это значение поступает в стек для вычисления выражения, определяющего значение at. Вычисляя, получаем at:=-1.667E-04;

4)выполняется оператор i:=2;

5)начинается исполнение оператора

while abs(at-ap)>eps do begin
  sum:=sum+at;

  ap:=at;

  at:=ap*znsin(i);

  i:=i+1

end;

Вычисляется условие abs(-1.667E-04 - 1.0E-01) > 1.0E-10 (получаем значение TRUE);

6) начинается исполнение тела цикла, выполняется оператор присваивания

sum:=sum+at;

получаем sum:=9.983E-02;

7) выполняется оператор присваивания

ap:=at;

получаем ap:=-1.667E-04;

8) выполняется оператор присваивания

at:=ap*znsin(i);

при вычислении выражения появляется динамический экземпляр данных функции znsin и передаются параметры (рис. 17.7)

	Znsin
	?
	Real

	i
	2
	Integer


Рис. 17.7. – Динамический экземпляр данных функции znsin
После этого начинается выполнение программного кода функции znsin.

Выполняется оператор присваивания

znsin:=-sqr(x)/((2*i)*(2*i+1));

вычисляем znsin:=-sqr(0.1)/((2*2)*(2*2+1)) (получаем -5.0Е-04);

функция znsin отработала; динамический экземпляр исчезает, а результат функции записывается в стек для вычисления at (at:=8.33E-8);

9) выполняется оператор присваивания

i:=i+1;

в результате чего i:=3;

10) вычисляется условие повторения цикла

abs(at-ap)>eps :

abs(8.33E-8 + 1.667E-4)>1.0E-10  (результат TRUE);

11) выполняется оператор присваивания  sum:=sum+at  :

sum:=9.983E-02 + 8.33E-08 (9.983E-02);

12) выполняется оператор присваивания ap:=at (ap:=8.33E-08);

13) выполняется оператор присваивания  at:=ap*znsin(i)  :

at:=8.33E-08 * znsin(3) (at:=8.33E-08 * -2.381E-04 = =- 1.984E-11);

14) выполняется оператор присваивания i:=i+1 : i:=3+1(4);

15) вычисляется условие повторения цикла

abs(at-ap)>eps :

abs( -1.984E-11 - 8.33E-08)>1.0E-10  (результат TRUE);

16) выполняется оператор присваивания  sum:=sum+at  :

sum:=9.983E-02 - 1.984E-11 (9.983E-02);

17) выполняется оператор присваивания ap:=at (ap:=-1.984E-11);

18) выполняется оператор присваивания  at:=ap*znsin(i)  :

at:=-1.984E-11 * znsin(4) (at:=-1.984E-11 * -1.389E-04 = =2.756E-15);

19) выполняется оператор присваивания i:=i+1 : i:=4+1(5);

20) вычисляется условие повторения цикла

abs(at-ap)>eps :

abs( 2.756E-15 + 1.984E-11)>1.0E-10  (результат FALSE);

оператор цикла прекращает выполнение;

21) выполняется оператор присваивания z6:=sum;
в результате переменная z6 получает значение 9.9833416647E-02.

Функция z6 выполнилась, динамический экземпляр функции исчезает.

Таким образом, значение 9.9833416647E-02 выступает четвертым параметром при обращении к процедуре вывода:

writeln('sin(',x,')=',z6(E,x,znsin))  .

В результате выполнения на экране имеем текст:

sin(1.0000000000E-01)=9.9833416647E-02

Программа fsin выполнилась за 5 шагов. 

